

## Electronic Supplementary Information (ESI)

# Zero-quantum filtered pure shift TOCSY

Jari J. Koivisto\*

*Department of Chemistry, Aalto University School of Chemical Technology, P.O. Box 16100,  
00076, Finland. E-mail: [jari.koivisto@aalto.fi](mailto:jari.koivisto@aalto.fi); Tel: +358 50 3438574*


### Table of contents

Experimental	S2
Molecular structure of (–)-eburnamonine	S2
Pulse sequence timing diagrams for the non-ZQF pure shift TOCSY and conventional ZQF TOCSY experiments	S3
Full conventional and pure shift TOCSY spectra	S4
Pulse sequence for the ZQF pure shift TOCSY experiment (Bruker spectrometers)	S7

## Experimental

All TOCSY experiments were performed in  $\text{CDCl}_3$  at 25 °C on a Bruker Avance DPX400 spectrometer equipped with a 5 mm BBFO probehead. A 20 ms DIPSI-2 sequence was used for the isotropic mixing. The conventional phase sensitive ZQF TOCSY spectrum of (–)-eburnamonine was recorded using a spectral width of 3491.62 Hz in both dimensions, 2 k data points, 256  $t_1$  increments, and 8 transients. The experimental time was 1 h 23 min. The data was zero-filled to a 2 k × 2 k matrix before Fourier transformation. The phase sensitive pure shift TOCSY spectra of (–)-eburnamonine were recorded using 2 k data points, 32  $t_2$  increments and 128  $t_1$  increments, and 4 transients. The experimental times were 11 h 34 min (non-ZQF version) and 11 h 59 min (ZQF version). Spectral widths were 3491.62, 99.761, and 3491.62 Hz in  $F_3$ ,  $F_2$ , and  $F_1$  dimensions, respectively. The pure shift spectra were recorded with the same receiver gain setting. The Bruker AU program “pshift” (modified to cope with the aqseq 321 data) was used to generate a 2D data set in which the  $F_2$  dimension was decoupled. The data was zero-filled to a 2 k × 2 k matrix before covariance processing (the Bruker AU program “covariance”). All data processing was made using TopSpin 3.0 software.


## Molecular structure of (–)-eburnamonine




**Scheme S1** Molecular structure of the indole alkaloid (–)-eburnamonine. Proton assignments are shown in red. The assigned  $^1\text{H}$  spectrum is shown in Fig. S2 ( $F_2$  projection).

## Pulse sequence timing diagrams for the non-ZQF pure shift TOCSY and conventional ZQF TOCSY experiments

(a)




(b)




**Fig. S1** Pulse sequences for the pure shift TOCSY experiment without zero-quantum filtering (a) and for the conventional ZQF TOCSY experiment (b).


## Full conventional and pure shift TOCSY spectra



**Fig. S2** Conventional ZQF TOCSY spectrum of (-)-eburnamonine. Projections: conventional  $^1\text{H}$  spectra. Proton assignments are given in red ( $F2$  projection).



**Fig. S3** Non-ZQF pure shift TOCSY spectrum of (-)-eburnamonine. Projections: 1D <sup>1</sup>H pure shift spectra.<sup>1</sup>



**Fig. S4** ZQF pure shift TOCSY spectrum of (-)-eburnamonine. Projections: 1D  $^1\text{H}$  pure shift spectra.<sup>1</sup>

- 1 J. A. Aguilar, S. Faulkner, M. Nilsson and G. A. Morris, *Angew. Chem., Int. Ed.*, 2010, **49**, 3901-3903.

## Pulse sequence for the ZQF pure shift TOCSY experiment (Bruker spectrometers)

```
;push2dzs_zqf_tocsy  
;  
;Zangger-Sterk pure shift sequence  
;2D homonuclear Hartman-Hahn transfer using DIPSI-2 sequence for mixing  
;with gradient/rf spoil pulses (ZQ-filter)  
;phase sensitive  
;  
;Jari J. Koivisto, Aalto University (12th September 2012)  
;  
;Morris, G. A.; Aguilar, J. A.; Evans, R.; Haiber, S.; Nilsson, M. J. Am. Chem. Soc. 2010, 132, 12770-12772.  
;Aguilar, J. A.; Faulkner, S.; Nilsson, M.; Morris, G. A. Angew. Chem., Int. Ed. 2010, 49, 3901-3903.  
;Thrippleton, M. J.; Keeler, J. Angew. Chem., Int. Ed. 2003, 42, 3938-3941.
```

;Data can be reconstructed using the 'pshift' macro (mod. by JJK to cope with the aqseq 321 data).

```
;$CLASS=HighRes  
;$DIM=3D  
;$TYPE=  
;$SUBTYPE=  
;$COMMENT=
```

```
#include <Avance.incl>  
#include <Grad.incl>  
#include <Delay.incl>
```

```
define delay tauA  
define delay tauB  
define delay tauC
```

```
;TOCSY:  
"d12=20u"  
"in0=inf1"  
"d0=in0/2-p1*4/3.1416"  
"FACTOR1=(d9/(p6*115.112))/2"  
"l1=FACTOR1*2"  
;ZS:  
"in10=inf2/2"  
"p2=p1*2"  
"tauA=in10/2-p16-d16"  
"tauB=in10-p16-d16-300u"  
"tauC=in10/2-p16-d16-350u-(dw*2)-(dw*2*cnst4)-de"
```

aqseq 321

```
1 ze  
2 d1  
3 d12 pl1:f1  
50u UNBLKGRAD  
p1 ph1  
d0  
p1 ph2  
10u gron0 ;start ZQF 1  
(p32:sp29 ph7):f1 ;SmoothedChirp  
d16 groff ;end ZQF 1  
d12 pl10:f1 ;power level for DIPSI-2  
4 p6*3.556 ph23 ;start TOCSY spinlock (DIPSI-2)  
p6*4.556 ph25  
p6*3.222 ph23  
p6*3.167 ph25  
p6*0.333 ph23  
p6*2.722 ph25  
p6*4.167 ph23  
p6*2.944 ph25  
p6*4.111 ph23  
p6*3.556 ph25  
p6*4.556 ph23
```

```
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25
p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25
p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 4 times l1          ;end TOCSY spinlock (DIPSI-2)
p31:gp1                  ;crusher gradient
10u gron0                 ;start ZQF 2
(p34:sp30 ph7):f1        ;SmoothedChirp
20u groff                 ;end ZQF 2
d31                       ;recovery delay
d12 p1:f1
p1 ph3                    ;end TOCSY
d10                       ;start ZS, incremented delay
tauA
p16:gp1*0.5              ;gradient +0.5
d16
p2 ph8                    ;hard 180
tauB
p16:gp1*-0.5             ;gradient -0.5
d16
300u gron2                ;slice selection gradient on
p12:sp2:f1 ph9:r         ;soft 180 (Rsnob.1000)
100u groff                ;slice selection gradient off
200u
p16:gp1*-1.0             ;gradient -1
d16
50u BLKGRAD
tauC
d10                       ;incremented delay
5 go=2 ph31
30u
d1 mc #0 to 2
    F1PH(rd10 & ip1, id0)
    F2QF(id10)
exit

;TOCSY
ph1=0 2                   ;1st hard 90
ph2=0 0 0 0 2 2 2 2     ;2nd hard 90
ph3=0 0 0 0 2 2 2 2     ;3rd hard 90
ph7=0                     ;ZQF 1 + 2
ph23=3                    ;DIPSI-2
ph25=1                    ;DIPSI-2
;ZS:
ph8= 0 0 0 0 0 0 0 0     ;hard 180
ph9= 0 0 1 1 0 0 1 1     ;soft 180
ph31=0 2 2 0 1 3 3 1     ;receiver

;Processing
;PHC0(F1): 90
;PHC1(F1): -180
;FCOR(F1): 1
```